An algorithm to apply Fisher's exact test to any two-dimensional contingency table



There are two main difficulties in applying Fisher's exact test to contingency tables with more than two rows and two columns. Taken together, these two difficulties have led some textbook authors to say categorically that the test can be used only for 2 ( 2 tables. The first difficulty is the lack of a simple algorithm that is known to be logically sound. The second difficulty is that the computation becomes extremely time-consuming when the table increases in size. This paper presents a solution to the first difficulty, and an approximate method of estimating the magnitude of the second difficulty.



Consider the following 3 ( 4 contingency table, which shows how many students (randomly sampled from a fictitious population) fall into four mutually exclusive subject categories and three mutually exclusive voting categories.



�Arts�Science�Divinity�Maths�Total��Party 1�3�1�0�6�10��Party 2�1�4�4�2�11��Party 3�0�1�5�0�6��Total�4�6�9�8�27��

Given the marginal totals, and the null hypothesis that the row categories are independent of the column categories, the expected frequencies within the table would be as follows.



�1.48�2.22�3.33�2.96���1.63�2.44�3.67�3.26���0.89�1.33�2.00�1.78��

The observed departure from the expected frequencies could be measured, but not reliably tested, by Pearson's (2 statistic, which in this example has a value of 17.675. 



In principle, Fisher’s method is quite simple. We generate every possible table that is compatible with the given marginal totals, and calculate the exact probability p of each table, using Fisher’s formula (1934). In the above example there are 8634 possible tables that would give the same marginal totals. We then need to partition the tables into those that are more extreme than the observed table (i.e. less compatible with the null hypothesis) and those that are less extreme. By summing the probabilities of the extreme tables we obtain a P value that is used in the usual classical way as a test of the null hypothesis.



One obvious approach is to calculate the exact probability that, given these marginal totals, the departure from the expected frequencies, as measured by (2 , would be equal to or greater than 17.675.



An alternative and intuitively appealing approach suggested by Anscombe ( 1981) was that the value of p calculated by Fisher’s formula for a particular possible table should be used not only for calculating the sum P of the probabilities of extreme tables, but also as a measure of how extreme the table is. Those values of p that are smaller than the p value of the observed table are summed to find the total probability of all tables more extreme than the observed one.  



A further detail remains to be decided: what do we do with tables that are exactly as extreme as the observed table? I have adopted another of Anscome’s suggestions here: the P value used to test Ho is the sum of the p values smaller than that of the observed table plus half the sum of the p values equal to that of the observed table. This decision has been the subject of a number of studies, so I shall not elaborate on it here. 



For contingency tables more complex than 2 ( 2 tables, the difficulty is to design a systematic way of generating every possible table, so that it can be tested for extremity and its exact probability calculated.



The systematic method developed here is to attach a unique numerical value L to each possible table, and work through the tables in order of their values of L, so that none of the tables can be overlooked or counted twice. Consider the cells of the table, i.e. the elements of an array A, to be read like a book from left to right and from top to bottom. These elements when read in that order represent the digits of the number L, (using arithmetic to any base that is greater than the largest element). When we are considering a particular element, we call the preceding elements the higher-order elements of the array, and the later numbers the lower-order elements. We start with the table that has the smallest possible value of L, and then repeatedly increment L by the smallest possible increment consistent with the constraints of the given marginal totals, and making sure that no element is ever less than zero. This algorithm depends on two subroutines described below.



The first subroutine (labelled as Minimise in the source code) minimises the value of the low-order elements of L, starting from row y and column x, without altering the higher-order elements, and without altering the marginal totals.



This subroutine depends on the following theorem, whose proof is rather too obvious to be worth spelling out:

the element A(y,x) can be reduced without altering the higher-order elements or the marginal totals if and only if there is a non-zero element A(y+v, x+u) where u, v are positive integers.



The second subroutine (labelled as Increment in the source code) is designed to increment by 1 the lowest-order element of A that is capable of being incremented without altering the higher-order elements or the marginal totals.



This subroutine depends on the following theorem, whose proof is even more obvious:

an element A(y,x) can be incremented without altering the higher-order elements or the marginal totals if and only if there exist two non-zero elements A(y+v,x) and A(y,x+u) where u, v and positive integers.



The subroutines Increment and Minimise are used alternately (in the UK sense of the word).



There are reasonable grounds for believing that this systematic method of working through the possible contingency tables does, in fact, generate every possible table exactly once. Firstly, the program based on it has given the same answers as other software packages when applied to 2 ( 2 tables. Secondly, it agrees exactly with the published analyses of Yates (1934) for 2 ( 3 tables. Thirdly, when the probabilities calculated by equation 1 are summed for all possible contingency tables consistent with a given set of marginal totals, the answer found has invariably been between 0.99999 and 1.00001 for numerous analyses in which the number of possible tables varies from 6 to 2954315. Fourthly, I have made the source code available so that users can check its logic (see http://www.statistics.org.uk).



For each of the possible contingency tables generated by the subroutine Increment, the calculation of its exact probability p is fairly simple, and is based on the work of Fisher (1934) and Yates (1934). If n1, n2, n3 . . . are the row totals and m1, m2, m3  . . . are the column totals and a1, a2, a3 . . . are the elements of the array, and G is the grand total, then



�EMBED Equation.3���

Apart from the difficulty of logical design, there remains the difficulty of lengthy computation. One method that I have used to reduce the computation time is to create a look-up table of logarithms of factorials at the start, so that the above formula can be evaluated quickly. Another obvious method is to calculate the logarithm of �EMBED Equation.3��� only once at the start of the calculation, because it is a common factor in the value of p for all possible contingency tables with the given marginal totals.



It is wise before embarking on a project to try to count the cost. If we started a Fisher’s exact test with no estimate of the time required, we could be in the tantalising position of having run the program for several days and be reluctant to abandon the attempt since the solution may be only minutes away. Our decision would be easier if we had an approximate measure of the number of possible tables that would be compatible with the marginal totals.



Since all the possible tables with D degrees of freedom could in principle be arranged in a D-dimensional array with diameters constrained by the marginal totals, one would expect that if D varied, and if the size of the elements were altered in proportion to a factor K, then the number of possible tables should change roughly in proportion to KD. The main constraints on the number of possible tables will be imposed by the smaller marginal totals, and should be relatively unaffected by the largest marginal totals. A small marginal total m belonging to a column with r elements will exert a constraint on r -1 independent variables. Each of these r -1 elements can have at most m +1 different values, ranging from 0 to m. These r -1 variables will therefore, at first sight, be expected to generate a possible (r - 1)m + 1 different sets of values. 



However, a further constraint may now be considered. If two independent variables are constrained by the same marginal total then they obviously cannot both have their maximum value at once. A table of possible sets of values would not in fact fill a square array with its two dimensions represented by the two variables: about half of the array would be inconsistent with the marginal total in question, and the remaining part of the array would be triangular. We should therefore divide the estimated number of tables by 2. Likewise, if there are three so-called independent variables constrained by the same marginal total, the expected cubic array of sets of values would be truncated to a triangular pyramid, and we should therefore divide the expected number of tables by 6, since the triangular pyramid will have one sixth of the volume of the cube that encloses it. In practice it is empirically found that this compensation for truncation gives an underestimate of the possible number of tables. A better estimate is usually obtained if we divide by 2 for every pair of variables constrained by the same marginal total, by 3 for every triplet and so on. 



The above argument is obviously very far from exact or rigorous, and is intended only to give some insight into the form of a relationship that might intuitively be expected between the marginal totals, the degrees of freedom and the number of possible contingency tables. The actual relationship used must ultimately be judged by empirical tests. 



The approximate algorithm that I have used to estimate the number of possible contingency tables is as follows. Write down all the r + c marginal totals (row and column totals) in ascending numerical order, writing each row total (c - 1) times and each column total (r - 1) times, keeping a note of which ones are row totals and which ones column totals. If there are more rows than columns, and one of the row totals is equal to one of the column totals, then give priority to the row total in forming the list, or vice versa. When the list contains exactly (r - 1)(c - 1) numbers, stop adding to it. Then increase each number by 1 and calculate the product obtained by multiplying all of them together. Finally divide the product by 2 for every marginal total that has been used twice in the calculation, of by 3 for every one that has been used three times, and so on. The answer thereby obtained usually approximates to the total number of possible tables, to within a factor of 2, and can be used to test the feasibility of a particular calculation before starting it. For programmers, the answer can be used to advise the software user of the estimated time of completion of the calculation.



�Table 1. Some examples of contingency tables analysed by the Exact Test. All these tables took less than 1 second of computing time with a 1.8 GHz Pentium 4 processor, except the last table, which took 62 seconds.



Observed contingency table�Estimated number

of possible tables

with same

marginal totals�Actual number of

possible tables

with same

marginal totals�P value

by Fisher’s

exact test��

��������25�1�������18�4���6�6�0.109����������50�4�12������2�2�3��56�35�0.017����������50�4�2������2�3�5��64�54�7.3 ( 10-6����������6�9�2������7�2�4��84�74�0.101����������25�8�14������55�27�9��864�798�0.0098����������1�6�0������0�12�3������2�6�1��1736�1224�0.157��1�14�0��������������0�7�0������0�11�4������3�6�0��1736�1224�0.011��1�14�0��������������3�1�0�6�����1�4�4�2�7503�8634�0.0033��0�1�5�0�������������3�1�0�6�����1�4�4�2�����0�1�5�0�5799660�2954315�0.020��2�5�3�6�������������

As the number of rows and columns increases, and as the counts within the cells increase, the calculation soon becomes impracticably slow. But it may be noticed that it is a calculation that lends itself very easily to vector processing or parallel processing, since there is no need to complete the analysis of one possible contingency table before beginning the analysis of other possible tables with the same marginal totals. We may expect therefore that when parallel and vector processors become readily available it should become easier to analyse large contingency tables in this way.

A Faster algorithm



When only part of a possible contingency table has been filled in, it is sometimes possible to recognise that the table when completed is bound to be extreme, or is bound not to be extreme. It is also possible to calculate the exact probability of a partly specified table. So it is not necessary to calculate the exact probability of every particular table: a whole set of tables can often be processed in one short calculation.



How do we calculate the exact probability of a partly specified table? The following procedure is designed only for tables that are filled in from left to right and from top to bottom, like writing a paragraph. The empty rows are all at the bottom, and the empty cells in the partly completed row are all to the right.



First, complete any row or column that has only one cell unspecified: this can be done in only one way that is compatible with the marginal totals.



Then, starting from the partly completed contingency table T0, construct two complete contingency tables T1 and T2 as follows. T1 has the same column totals as T0 and comprises all the complete rows of T0, plus one complementary row, which can be filled in in only one way that is compatible with the column totals. 



Construct T2 from the partly filled row of T0 plus one complementary row. The column totals of T2 are equal to the contents of the last row of T1. 



At this stage, the rationale of the calculation begins to be clear. The hypothetical allocation of numbers in all possible ways to complete the table T0 could be done in two stages, with calculable probabilities. First, the numbers could be allocated to the cells of T1, and then the last row of T1 could be further divided among the cells of T2. The exact probability of T0 is therefore the product of the probabilities of T1 and T2. 



The exact probability p1 of T1 is calculated using Fisher’s formula. Before doing the same with T2, the empty columns to the right of T2 are merged into one empty column, whose contents can be filled in in only one way that is compatible with the marginal totals. We can then calculate the exact probability p2 of T2 and hence calculate p = p1 ( p2, which is the exact probability of the partly specified table T0.



A worked example is shown below. Suppose our partly completed table T0 is as follows.



3�1�2�5�11��4�6���20������8������4������6��12�11�14�12�49��

Then T1 is as follows, and its exact probability p1 is 0.0059862.



3�1�2�5�11��9�10�12�7�38��12�11�14�12�49��

Table T2 is initially 

4�6�?�?�20��5�4�?�?�18��9�10�12�7�38��

 and when condensed it becomes 



4�6�10�20��5�4�9�18��9�10�19�38��

And its exact probability is 0.0727953.



The exact probability of table T0 is therefore 0.0059862 ( 0.0727953 = 0.000435767.



One special case needs to be catered for explicitly: if table T0 has no complete rows, then T1 is empty and p1 = 1. 



How do we recognise that table T0 is bound to lead only to extreme tables? 



Suppose we are using the (2 statistic as a measure of how extreme a possible table is. 

The contribution of the specified cells to the total of the (2 statistic is easily calculated. Also, it is shown below that a lower limit for the possible contribution of the other cells can easily be calculated. 



Theorem

If there is a set of cells with expected frequencies adding up to a total T, and there are N observations to be distributed among these cells, then the contribution of these cells to (2 is greater than or equal to �EMBED Equation.3���.



Proof

Suppose that the observed counts Oi are distributed in proportion to the expected frequencies Ei , i.e. �EMBED Equation.3���. Then suppose we try to reduce the value of (2 by increasing one particular observation Oi by a positive amount (, at the expense of reducing another observation Oj by the same amount. In practice, if the observed counts are all integral ( = 1, but this is not essential to the proof. Then the resulting increase in (2 will be �EMBED Equation.3���

But we know that �EMBED Equation.3���, so the increase in (2 is �EMBED Equation.3���, which is always positive. Fine adjustments to the observed frequencies can only increase (2 , so we know that the value of (2 is at a minimum. Its actual value is �EMBED Equation.3���



The algorithm that I have used applies this lower limit to each column of the contingency table in turn, or at least to that part of the column that has not yet been specified.



�For example, suppose we have the partly specified table as shown below.

3�1�2�5�11��4�6���20������8������4������6��12�11�14�12�49��

Then the number of observations still to be allocated is N = 49 – 21 = 28.



The expected frequencies (before any cells were allocated) are 



2.69�2.47�3.14�2.69��4.90�4.49�5.71�4.90��1.96�1.80�2.29�1.96��0.98�0.90�1.14�0.98��1.47�1.35�1.71�1.47��

The column totals of the original expected frequencies of the unspecified cells are

4.41�4.05�10.85�9.31��

And the parts of the observed column totals still to be allocated are 

5�4�12�7��

Then we know that the first column of unspecified cells must contribute at least 

�EMBED Equation.3��� to the total value of (2 , and the other columns must contribute at least 0.000617, 0.121889 and 0.573158 respectively. Given that the specified cells already contribute 3.972, the total value of (2 for the whole table must be at least 4.74656.



Now how do we find an upper limit for the value of (2?



If a given row has a total of N observations still available to be distributed among its blank cells, then find the cell with the smallest expected frequency E and fill it to the limit imposed by its column total. Then fill the cell with the next smallest E similarly, and so on until all N observations have been distributed. The value of (2 contributed by that row is then at least as large as it theoretically can be. 



For example, suppose we have the partly specified table as shown below.

3�1�2�5�11��4�6���20������8������4������6��12�11�14�12�49��

The expected frequencies (before any cells were allocated) are 



2.69�2.47�3.14�2.69��4.90�4.49�5.71�4.90��1.96�1.80�2.29�1.96��0.98�0.90�1.14�0.98��1.47�1.35�1.71�1.47��

Then the number of observations still to be allocated in rows 2, 3, 4 and 5 is N = 10, 8, 4 and 6 respectively. Distributing these totals in accordance with the above rule as if all rows were mutually independent gives 



3�1�2�5�11��4�6�3�7�20��4�4�0�0�8��0�4�0�0�4��2�4�0�0�6��12�11�14�12�49��

But this does not in fact maximise (2 for each row. This gives (2 = 37.66, but changing the allocation as follows increases (2 to 61.97.



3�1�2�5�11��4�6�10�0�20��0�1�0�7�8��0�4�0�0�4��0�0�0�6�6��12�11�14�12�49��

Suppose we were using Fisher’s p as a measure of how extreme a table is. 



�EMBED Equation.3���

This expression can be factorised into �EMBED Equation.3���and a set of separate factors �EMBED Equation.3���for each row or column. For a given row or column with c cells still to be allocated and a total of N observations to be distributed among them, the minimum conceivable value of p will occur when a1 = a2 = a3 . . . = N/c and the maximum conceivable value of p will occur when one of the cells contains N and the others contain zero. Thus the contribution of a

particular row or column to ln(p) will always lie between �EMBED Equation.3���and -�EMBED Equation.3���



If the column totals vary more than the row totals, this formula is best applied column by column; otherwise it is best applied row by row. This will help to make the limits narrower and more informative. 



For example, suppose we have the partly specified table as shown below.

3�1�2�5�11��4�6���20������8������4������6��12�11�14�12�49��

Rows 2, 3, 4 and 5 respectively have 10, 8, 4 and 6 observations still to be distributed. The contributions of these rows to ln(p) therefore lies between the following values.



Row no.�N�c�N/c�cLn((N/c)!) = Min

contribution to ln(p)�Ln(N!) = Max

contribution to ln(p)��2�10�2�5�9.574983�15.10441��3�8�4�2�2.772589�10.6046��4�4�4�1�0�3.178054��5�6�4�1.5�5.3174�6.579251��Total����17.66497�35.46632��



The contribution U of the unspecified cells to ln(p) is therefore somewhere between 

17.66497 and 35.46632.



The contribution S of the six specified cells to ln(p) is �EMBED Equation.3��� We also use the fact that the contribution M of the marginal totals is  �EMBED Equation.3���18.30205

The conclusion is that �EMBED Equation.3���

and so p = 1.412 ( 10-15 to 7.599 ( 10-8. 



If the value of Fisher’s p for the observed table lies outside this range then we do not need to construct all possible tables that start with 3, 1, 2, 5, 4, 6 . . . because we know whether they are more extreme or less extreme than the observed table. We merely need to calculate the sum of their p values, and this can be done very quickly. It is the same as the a priori probability that the table will start with 3, 1, 2, 5, 4, 6.





How do we calculate the exact probability of a partly specified table? 

The following procedure is designed only for tables that are filled in from left to right and from top to bottom, like writing a paragraph. The empty rows are all at the bottom, and the empty cells in the partly completed row are all to the right.

First, complete any row or column that has only one cell unspecified: this can be done in only one way that is compatible with the marginal totals.

Then, starting from the partly completed contingency table T0, construct two complete contingency tables T1 and T2 as follows. T1 has the same column totals as T0 and comprises all the complete rows of T0, plus one complementary row, which can be filled in in only one way that is compatible with the column totals. 

Construct T2 from the partly filled row of T0 plus one complementary row. The column totals of T2 are equal to the contents of the last row of T1. 

At this stage, the rationale of the calculation begins to be clear. The hypothetical allocation of numbers in all possible ways to complete the table T0 could be done in two stages, with calculable probabilities. First, the numbers could be allocated to the cells of T1, and then the last row of T1 could be further divided among the cells of T2. The exact probability of T0 is therefore the product of the probabilities of T1 and T2. 

The exact probability p1 of T1 is calculated using Fisher’s formula. Before doing the same with T2, the empty columns to the right of T2 are merged into one empty column, whose contents can be filled in in only one way that is compatible with the marginal totals. We can then calculate the exact probability p2 of T2 and hence calculate p = p1 ´ p2, which is the exact probability of the partly specified table T0.

A worked example is shown below. Suppose our partly completed table T0 is as follows.

3�1�2�5�11��4�6���20������8������4������6��12�11�14�12�49��Then table T1 is as follows, and its exact probability p1 is 0.0059862.

3�1�2�5�11��9�10�12�7�38��12�11�14�12�49��Table T2 is initially 

4�6�?�?�20��5�4�?�?�18��9�10�12�7�38�� and when condensed it becomes 

4�6�10�20��5�4�9�18��9�10�19�38��And its exact probability is 0.0727953.

The exact probability of table T0 is therefore 0.0059862 ´ 0.0727953 = 0.000435767.

One special case needs to be catered for explicitly: if table T0 has no complete rows, then T1 is empty and p1 = 1. 



Counting the Cost



It is wise before embarking on a calculation to try to count the cost. If we started a Fisher’s exact test with no estimate of the time required, we could be in the tantalising position of having run the program for several days and be reluctant to abandon the attempt since the solution may be only minutes away. Our decision would be easier if we had an approximate measure of the number of possible tables that would be compatible with the marginal totals.

Gail and Mantel proposed an approximate method of estimating the total number of possible tables that are compatible with given marginal totals. Their algorithm has been used by Metha and Patel in their calculation of exact probabilities. I have used a similar but not identical method. My software agrees with the worked example of Gail and Mantel, but does not agree with the calculations of Metha and Patel: I suspect that they have made a programming error. 

A slight but useful refinement can be made to Gail and Mantel’s method. Their formulae do not treat rows and columns as equivalent. The predicted number of possible tables changes when the table is rotated or transposed so that the rows become columns and the columns become rows. I have found that the prediction is substantially better if Gail and Mantel’s method is carried out twice, rotating the table between runs, and then using the mean of the two estimates. See fig. 1.

As the number of rows and columns increases, and as the counts within the cells increase, the calculation soon becomes impracticably slow. But it may be noticed that it is a calculation that lends itself very easily to vector processing or parallel processing, since there is no need to complete the analysis of one possible contingency table before beginning the analysis of other possible tables with the same marginal totals. We may expect therefore that when parallel and vector processors become readily available it should become easier to analyse large contingency tables in this way.

�Appendix: Visual Basic Source Code of Fisher.exe



DefInt A, C-D, I-J, R, X-Y: Rem this cuts execution time by a factor of 13/23

Dim rows, cols, L1, L2, P, P2, PT, LF(), detailed, allpvals, Pobs, Lobs, Peq, T1, T2

Dim spacing1, spacing2, spacing3, spacing4, changing, tolerance



Private Sub Command1_Click()

cols = Int(Val(Text1.Text))

rows = Int(Val(text2.Text))



label9.Caption = " "

label10.Caption = " "

label11.Caption = " "

label12.Caption = " "



allpvals = 0: Rem -1 specifies that all P values of possible tables are to be listed

detailed = 0: Rem -1 specifies that a detailed table-by-table report is wanted

ReDim n(rows), m(cols), ob(rows, cols), ex(rows, cols), A(rows, cols)



Open "fisher.prn" For Output As 2

If detailed Or allpvals Then Open "detail.prn" For Output As 3

Print #2, rows; "rows and "; cols; "columns"

If detailed Then Print #3, "Observed frequencies:"

Print #2, "Observed Frequencies:"



Rem read the data

g = 0: Rem Grand total of cell contents

For i = 0 To rows - 1: grid1.Row = i

For j = 0 To cols - 1: grid1.Col = j

x = Val(grid1.Text)

ob(i + 1, j + 1) = x: A(i + 1, j + 1) = x: n(i + 1) = n(i + 1) + x: m(j + 1) = m(j + 1) + x: g = g + x

Print #2, Format$(x, "  ####.##");: If detailed Then Print #3, x;

Next j: Print #2,: If detailed Then Print #3,

Next i



rechecking = False

recheck:

'Check for any empty rows

condensed = False

i = 1

nextrow:

If n(i) = 0 Then

If Not (condensed Or rechecking) Then MsgBox "Condensing table to remove empty rows and/or columns"

If i < rows Then

    For k = i To rows - 1: n(k) = n(k + 1)

    For j = 1 To cols: ob(k, j) = ob(k + 1, j): A(k, j) = A(k + 1, j)

    Next j: Next k: n(k) = 0

End If

rows = rows - 1: If rows > 1 Then vscroll2.Value = rows

condensed = True

End If

i = i + 1: If i < rows + 1 Then GoTo nextrow



'Check for any empty columns

i = 1

nextcolumn:

If m(i) = 0 Then

If Not (condensed Or rechecking) Then MsgBox "Condensing table to remove empty column(s)"

If i < cols Then

    For k = i To cols - 1: m(k) = m(k + 1)

    For j = 1 To rows: ob(j, k) = ob(j, k + 1): A(j, k) = A(j, k + 1)

    Next j: Next k: m(k) = 0

End If

cols = cols - 1: If cols > 1 Then vscroll1.Value = cols

condensed = True

End If

i = i + 1: If i < cols + 1 Then GoTo nextcolumn



'Update grid if necessary

If condensed Then

For i = 0 To rows - 1: grid1.Row = i

For j = 0 To cols - 1: grid1.Col = j

grid1.Text = Str$(ob(i + 1, j + 1))

Next j: Next i

End If



If condensed And (rows > 2 Or cols > 2) Then rechecking = True: GoTo recheck 'Make sure there are no more empty rows or columns



If rows < 2 Or cols < 2 Then MsgBox "There must be at least two rows and two columns in the contingency table": GoTo skipcalc



Form2.Hide

Cls



Rem Now calculate and print expected frequencies and observed chisquare

Rem and the exact probability of the observed table, for use in recognising

Rem more extreme tables.

Print #2,: Print #2, "Expected frequencies:"

If detailed Then Print #3, "Expected frequencies:"

For Row = 1 To rows: For Col = 1 To cols

ex(Row, Col) = n(Row) * m(Col) / g: Print #2, Format$(ex(Row, Col), "  ####.##");: If detailed Then Print #3, ex(Row, Col);

Pchis = Pchis + (ob(Row, Col) - ex(Row, Col)) ^ 2 / ex(Row, Col)

Rem Pchis is Pearson's chi-squared; Pchisy is same with Yates' correction

Pchisy = Pchisy + (Abs(ob(Row, Col) - ex(Row, Col)) - 0.5) ^ 2 / ex(Row, Col)

Next Col: Print #2,: If detailed Then Print #3,

Next Row: Print #2,: If detailed Then Print #3,: Print #3,:

Print #2, "Chi squared ="; Pchis; "or, with Yates' correction,"; Pchisy

If detailed Then Print #3, "Chi squared ="; Pchis; "or, with Yates' correction,"; Pchisy



Rem Prepare a look-up table of logarithmic factorials to avoid repetitive

Rem calculation

ReDim LF(g)

For i = 1 To g: LF(i) = LF(i - 1) + Log(i): Next i

Rem and calculate the log, L1, of the factors in the P formula that are the

Rem same for all possible contingency tables with the given marginal totals.

L1 = 0

For i = 1 To rows: L1 = L1 + LF(n(i)): Next i

For i = 1 To cols: L1 = L1 + LF(m(i)): Next i

L1 = L1 - LF(g)



Rem Now calculate exact prob. of the observed table, so that more extreme

Rem tables can be recognised:

P = 0: Pobs = 1: Lobs = 0: Call hypergeometric(A()): Pobs = P: Lobs = L2

Peq = 0: Rem Total Probability of all tables of probability exactly Pobs.

Print #2, "Exact prob. of the observed table, given the marginal totals ="; P

If detailed Then Print #3,: Print #3, "Possible tables:"



Rem Estimate approximate no. of possible contingency tables that

Rem have the given marginal totals, and hence estimate the time required to

Rem complete the calculation

ReDim marg(rows + cols), edge(rows + cols): Rem Marg are marginal totals plus 1;

Rem edge indicates by the no. of rows or the no. of cols which edge of the

Rem table a particular marginal total was obtained from.

GoTo L3

subr:

For i = 1 To rows: marg(i) = n(i) + 1: edge(i) = cols - 1: Next i

For i = 1 To cols: marg(i + rows) = m(i) + 1: edge(i + rows) = rows - 1: Next i

Return

subc:

For i = 1 To cols: marg(i) = m(i) + 1: edge(i) = rows - 1: Next i

For i = 1 To rows: marg(i + cols) = n(i) + 1: edge(i + cols) = cols - 1: Next i

Return

L3:

If rows > cols Then GoSub subr Else GoSub subc

product = 1: factors = (rows - 1) * (cols - 1)

L4:

Min = marg(1): fact = edge(1): element = 1

For i = 2 To rows + cols: If marg(i) < Min Then Min = marg(i): fact = edge(i): element = i

Next i

marg(element) = g: Rem to make sure this element is not used again

uses = 0

L5:

product = product * Min: uses = uses + 1: factors = factors - 1

If factors > 0 And uses < fact Then GoTo L5

product = product / uses

If factors > 0 Then GoTo L4

product = Int(product)

Print #2, "Estimated no. of possible tables with same marginal totals ="; product

text3.Text = Str$(product)

If detailed Then Print #3, "Estimated no. of possible tables with same marginal totals ="; product



Rem Main part of program

T1 = Now: text5.Text = "calculating"

x = 1: y = 1: Rem Defines element of table at which minimisation of L is to start

GoSub minimise

tables = 1: Rem No. of possible tables so far analysed

P = 0: PT = 0

loop1:

If detailed Then GoSub printL

Call hypergeometric(A())

If detailed Or allpvals Then Print #3, "P for this table ="; P2

If tables / 1000 = Int(tables / 1000) Then GoSub progress

GoSub increment

increased = success

x = x + 1: If x = cols Then x = 1: y = y + 1

If increased Then tables = tables + 1: GoSub minimise: GoTo loop1



'Report the results

Pans = P + Peq / 2 ' P value as defined by Anscombe

Print #2,: Print #2, "Actual no. of tables with same marginal totals ="; tables

Print #2, "Probability of a more extreme table     = Pex ="; P

Print #2, "Probability of an equally extreme table = Peq ="; Peq

Print #2, "Probability defined by Anscombe = Pex + Peq/2 ="; Pans

If detailed Then Print #3, "Probability of a more extreme table    ="; P: Print #3, "Probability of an equally extreme table ="; Peq

Print #2, "Total P for all tables (should be 1)="; PT

GoSub progress

Print #2,: Print #2, "As a test of the null hypothesis that the row categories are independent of the"

Print #2, "column categories, the author recommends that you use the P value as defined by Anscombe,"

Print #2, "namely P ="; Pans

Print #2, "Earlier statisticians tended to use the P value giving the probability of an equally extreme"

Print #2, "or more extreme table, namely P ="; P + Peq

If Pans > 0.000001 Then text5.Text = Format$(Pans, "0.############") Else text5.Text = Str$(Pans)

Close

GoTo skipcalc



'Subroutine to update the display of the number of tables analysed, and time taken so far

progress:

text4.Text = Str$(tables): T2 = Now - T1

label9.Caption = Str$(Int(T2)) + " days": T2 = 24 * (T2 - Int(T2))

label10.Caption = Str$(Int(T2)) + " hours": T2 = 60 * (T2 - Int(T2))

label11.Caption = Str$(Int(T2)) + " minutes": T2 = 60 * (T2 - Int(T2))

label12.Caption = Str$(Int(T2)) + " seconds"

label12.Refresh

DoEvents 'To allow user to shrink Fisher to an icon or run other programs

Return



Rem Subroutine to increment by 1 the lowest-order element of A(,) that can be

Rem incremented. This method depends on the following theorem:

Rem an element A(y,x) can be incremented without altering any of the higher-

Rem order elements if and only if there exist two non-zero elements A(y+v,x)

Rem and A(y,x=u) where u, v > 0.

increment:

y = rows - 1: x = cols - 1: Rem Define the element that we first wish to try to increment

success = 0

start1:

For Row = rows To y + 1 Step -1

For Col = cols To x + 1 Step -1

If A(Row, x) = 0 Or A(y, Col) = 0 Then GoTo skip1

A(y, x) = A(y, x) + 1: A(Row, Col) = A(Row, Col) + 1

A(y, Col) = A(y, Col) - 1: A(Row, x) = A(Row, x) - 1

success = -1: Row = y - 1: Col = x - 1: Rem to avoid further looping

skip1:

Next Col: Next Row

If (x = 1 And y = 1) Or success Then Return

x = x - 1: If x = 0 Then x = cols: y = y - 1

GoTo start1



Rem Subroutine to minimise the value of the lower-order elements of L, starting

Rem from row y and column x, without affecting the higher-order elements.

Rem This method depends on the following theorem:

Rem The element A(y,x) can be reduced without altering any of the higher-order

Rem elements if and only if there exists a non-zero element A(y+v,x+u) where

Rem u, v > 0.

minimise:

start3:

If y = rows Or x = cols Then GoTo skip3

start2:

success = 0

For Row = rows To y + 1 Step -1

For Col = cols To x + 1 Step -1

If A(Row, Col) = 0 Then GoTo skip2

D = A(Row, Col): If A(y, x) < D Then D = A(y, x)

A(y, x) = A(y, x) - D: A(Row, Col) = A(Row, Col) - D

A(y, Col) = A(y, Col) + D: A(Row, x) = A(Row, x) + D

success = -1

skip2:

Next Col: Next Row

If success And A(y, x) > 0 Then GoTo start2

x = x + 1: If x = cols Then x = 1: y = y + 1

GoTo start3

skip3:

Return



Rem Subroutine to print to Detail.prn the elements of array A(,)

printL:

Print #3,

For i = 1 To rows: For j = 1 To cols: Print #3, A(i, j);: Next j: Print #3,: Next i

Return

skipcalc:

Close

End Sub



Private Sub Command2_Click()

End

End Sub



Private Sub Command3_Click()

If text5.Text = "calculating" Then GoTo skipcommand3

Open "fisher.prn" For Input As 3

While Not EOF(3)

    r$ = Input$(1, 3): r1$ = r1$ + r$

Wend

Form2.Show

Form2.Text1.Text = r1$

Close 3

skipcommand3:

End Sub



Private Sub Command4_Click()

Rem Clear the data

cols = Int(Val(Text1.Text))

rows = Int(Val(text2.Text))

ReDim n(rows), m(cols), ob(rows, cols), A(rows, cols)

g = 0: Rem Grand total of cell contents

For i = 0 To rows - 1: grid1.Row = i

For j = 0 To cols - 1: grid1.Col = j

grid1.Text = ""

ob(i + 1, j + 1) = 0: A(i + 1, j + 1) = 0: n(i + 1) = 0: m(j + 1) = 0

Next j: Next i

End Sub



Private Sub Form_Load()

spacing1 = Form1.Height - command1.Top

spacing2 = Form1.Height - frame2.Top

'spacing2 = form1.Height - label5.Top

'spacing3 = form1.Height - label6.Top

spacing4 = Form1.Height - text5.Top

tolerance = 0.0000000000001

'This determines how equally extreme two tables must be in order to be

'counted as equal. Tolerance = 0.01 to .00000000000001 gives the same answers.

End Sub



Private Sub Grid1_KeyUp(keycode As Integer, Shift As Integer)

If keycode > 47 And keycode < 58 Then

grid1.Text = grid1.Text + Chr$(keycode)

End If

If keycode > 95 And keycode < 106 Then

grid1.Text = grid1.Text + Chr$(keycode - 48)

End If

If keycode = 109 Or keycode = 189 Then grid1.Text = "-" & grid1.Text



If keycode = 8 And grid1.Text <> "" Then

grid1.Text = Left$(grid1.Text, Len(grid1.Text) - 1)

End If

If keycode = 46 Then grid1.Text = ""

'If grid1.Col = 1 Then x1$(grid1.Row) = grid1.Text

'If grid1.Col = 2 Then y1$(grid1.Row) = grid1.Text



End Sub



Private Sub hypergeometric(A())

Rem subroutine to calculate exact probability of this particular table

Rem and add it to the cumulative prob. of all tables more extreme than the

Rem observed table, i.e. more extreme as judged by its improbability.

L2 = L1

For Row = 1 To rows: For Col = 1 To cols: L2 = L2 - LF(A(Row, Col)): Next Col: Next Row

P2 = Exp(L2)

If detailed Then

    If Abs(L2 - Lobs) < tolerance And Pobs < 1 Then

        Print #3, "This is an equally extreme table "

    Else

        If L2 < Lobs And Pobs < 1 Then Print #3, "This is a table more extreme than the observed one"

    End If

End If

'If P2 < Pobs Then P = P + P2

'If P2 = Pobs Then Peq = Peq + P2

If Abs(L2 - Lobs) < tolerance Then

    Peq = Peq + P2

Else

    If L2 < Lobs Then P = P + P2

End If

PT = PT + P2

End Sub



Private Sub Label6_Click()



End Sub



Private Sub label7_Click()



End Sub



Private Sub Text1_Change()

cols = Int(Val(Text1.Text))

If cols > 1 Then

    grid1.Width = cols * 698 + 8

    grid1.cols = cols

Else

    cols = 2

End If

If cols > 20 Then cols = 20

If cols > 6 Then

Form1.Width = cols * 685 + 200

Else

Form1.Width = 4968

End If

If Not changing Then vscroll1.Value = cols

End Sub



Private Sub Text2_Change()

rows = Int(Val(text2.Text))

If rows > 1 Then

    grid1.Height = rows * 290 + 14

    grid1.rows = rows

Else

    rows = 2

End If

If rows > 20 Then rows = 20

If rows > 8 Then

Form1.Height = (rows + 14) * 285

Else

Form1.Height = 6264

End If

command1.Top = Form1.Height - spacing1

command2.Top = Form1.Height - spacing1

command3.Top = Form1.Height - spacing1

command4.Top = Form1.Height - spacing1

frame2.Top = Form1.Height - spacing2

text5.Top = Form1.Height - spacing4

label7.Top = Form1.Height - spacing4

If Not changing Then vscroll2.Value = rows

End Sub



Private Sub VScroll1_Change()

changing = True

Text1.Text = Str$(vscroll1.Value)

changing = False

End Sub



Private Sub VScroll2_Change()

changing = True

text2.Text = Str$(vscroll2.Value)

changing = False

End Sub

